

Belastingdienst

RegelSpraak-specificatie

_

syntaxdiagrammen

Datum: 24-1-2025

Versie 2.1.0

© 2024 Belastingdienst

Alle rechten voorbehouden.

Niets uit deze uitgave mag worden verveelvoudigd, opgeslagen in een geautomatiseerd gegevensbestand en/of openbaar gemaakt in enige vorm of op enige wijze, hetzij elektronisch, mechanisch, door fotokopieën, opnamen of op enige andere manier zonder voorafgaande schriftelijke toestemming van de auteur.

Inhoudsopgave:

Versiebeheer	
SyntaxdiagrammenSyntaxdiagrammen	5
13.2 Standaard syntax patronen	7
13.2.1 Digit	7
13.2.2 Getal	7
13.2.3 Geheel getal	7
13.2.4 Decimaal getal	7
13.2.5 Rationeel getal	8
13.2.6 Letter	8
13.2.7 Leesteken	9
13.2.8 Karakterreeks	9
13.2.9 Lidwoord	9
13.2.10 Bepaallidwoord	
13.2.11 Onbepaaldlidwoord	
13.2.13 Waarde	
13.2.13 Enumeratiewaarde	
13.2.14 Tekstwaarde	
13.2.15 Boolean waarde	
13.2.16 Getalwaarde	
13.2.17 Percentage	
13.2.18 De dato	
13.2.19 Datumwaarde	
13.2.20 Tijdwaarde	
13.2.21 Dag	
13.2.22 Maand	
13.2.23 Jaar	
13.2.24 Uur	
13.2.25 Minuut	
13.2.26 Seconde	
13.2.28 Naamwoord	
13.2.29 Naam	_
13.2.30 Meervoudsvorm	
13.3 Objecten en parameters	
13.3.1 Objecttypen	
13.3.2 Attributen en kenmerken	
13.3.3 Datatypen	
13.3.4 Domeinen	
13.3.5 Eenheden	
13.3.6 Tijdlijnen	
13.3.7 Dimensies	
13.3.8 Parameters	
13.3.9 Feittypen	
13.3.10 Dagsoort	
13.4 RegelSpraak	
13.4.1 Onderwerpketen	
13.4.2 RegelSpraak-regel	
13.4.3 Resultaatdeel	. 30

13.4.4 Gelijkstelling	31
13.4.5 Kenmerktoekenning	31
13.4.6 ObjectCreatie	31
13.4.7 FeitCreatie	32
13.4.8 Consistentieregels	32
13.4.9 Initialisatie	35
13.4.10 Verdeling	35
13.4.11 Dagsoortdefinitie	37
13.4.12 Voorwaardendeel	37
13.4.13 Samengestelde voorwaarde	39
13.4.14 Elementaire voorwaarde	41
13.4.15 Berekening	63
13.4.16 Expressie	

Versiebeheer

Onderstaande tabel bevat het overzicht van gepubliceerde versies. In de omschrijving staan de wijzigingen ten opzichte van de vorige versie.

Versie	Status	Datum	Omschrijving
1.00	Definitief	01-05-2023	Initiële versie.
1.0.1	Definitief	16-05-2023	Aanpassingen t.b.v. publicatie.
			Namen en informatie over concept-versies en mogelijke
			toekomstige aanpassingen verwijderd.
1.1.0	Definitief	24-10-2023	Par. 12.2.13 - Enumeratiewaarde als aparte literal expressie
			toegevoegd in verband met presentatie met enkele
			aanhalingstekens.
			Par. 12.3.3.7 - In getalspecificatie "reëel getal" gewijzigd in
			"getal".
			Par. 12.3.8 - Aanduiding "binair" verwijderd bij feittype.
			Par. 12.4.16.2 - Specificatie syntax Concatenatie expressie
			aangescherpt met gebruik "of".
			Par. 12.4.16.21/22/23 - Begrenzingexpressie en expressies
1 2 0	D - f: : t-: - f	44.04.2024	voor minimum en maximum begrenzing toegevoegd.
1.2.0	Definitief	11-04-2024	Specificatie van de expressie renteberekening verwijderd. Deservations de expressie van de de expressie van Besching verwijderd.
			Deze expressie maakt geen deel meer uit van RegelSpraak.
			Par. 12.4.16.14 Machtsverheffenfunctie toegevoegd.
			Par 12.4.16.17 Specificatie rekenkundige expressie "absolute waarde van" toegevoegd
			12.4.16.28 Alternatief "de absolute tijdsduur van"
			toegevoegd.
			Par. 12.4.16.40-49 Specificaties syntax van aggregatie
			gecorrigeerd.
2.0.0	Definitief	27-09-2024	Groot aantal aanvullingen op de syntax in verband met het
			toevoegen van tijdsafhankelijk rekenen.
2.1.0	Definitief	24-1-2025	Par. 13.4.2.11 Mogelijkheid van lidwoord bij naam variabele
			toegevoegd.
			Par 13.4.14.44/45 Mogelijkheid om "heeft" te gebruiken bij
			rolcheck toegevoegd.

Tabel 1

Syntaxdiagrammen

In hoofdstuk 13 van het RegelSpraak specificatie document is de RegelSpraak Syntax formeel vastgelegd met gebruik van de CORBA scripting taal. Om deze formele vastlegging inzichtelijker te maken, is ervoor gekozen om deze in dit document ook weer te geven in syntaxdiagrammen. Bij de opbouw van dit document is daarom gekozen voor het aanhouden van de hoofdstukstructuur en (sub)nummering zoals gebruikt in het RegelSpraak specificatie document.

Voor het genereren van syntaxdiagrammen uit syntax is gebruik gemaakt van de website Railroad Diagram Generator: https://rr.red-dove.com/ui. Deze website kan EBNF syntax omzetten in syntaxdiagrammen en is gebaseerd op de GitHub-pagina https://github.com/GuntherRademacher/rr. Voor de syntaxspecificatie van RegelSpraak is gekozen voor de notatiewijze die de OMG voor de CORBA scripting taal hanteert, waardoor een aantal transformaties nodig was om voor de website bruikbare input te leveren. Deze transformaties staan beschreven in onderstaande tabel. Bij de syntaxdiagrammen van paragraaf 13.3.1 zal ter illustratie de syntax, zoals die in dit document gebruikt wordt, worden weergegeven samen met de gewijzigde input die is ingegeven op de website.

Symbool	Is vervangen door
<	
>	
{	(
})
[(
])?
\n	"\n"
\t	"\t"

Tabel 1: benodigde transformaties van de syntax voor de generatie van syntaxdiagrammen.

Hoewel syntaxdiagrammen voor velen intuïtiever te lezen zijn dan de syntax zelf is een toelichting van de elementen die voorkomen in een syntaxdiagram van belang om eventuele misverstanden te voorkomen. Een legenda is daarom toegevoegd in onderstaande tabel.

Symbool/Weergave	Betekenis
>>	Begin van het diagram
→	Eind van het diagram
	Diagram gaat verder op de volgende regel
	Vervolg van het diagram op de vorige regel
- A -	Element bestaat uit karakter 'A'
- A	Element A is op een andere plek gedefinieerd




Tabel 2: Legenda syntaxdiagrammen.

13.2 Standaard syntax patronen


13.2.1 Digit

<digit> ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"




13.2.2 Getal

<getal> ::= <geheelgetal> | <decimaalgetal> | <rationeelgetal>




13.2.3 Geheel getal

<geheelgetal> ::= ["-"]<digit>+



13.2.4 Decimaal getal

<decimaalgetal> ::= ["-"]<digit>+ "," <digit>+




13.2.5 Rationeel getal

<rationeelgetal> ::= <geheelgetal>["_"<geheelgetal>]"/"<geheelgetal>




13.2.6 Letter

<letter> ::= "a" | "b" | "c" | "d" | "e" | "f" | "g" | "h" | "i" | "j" | "k" | "l" | "m" | "o" | "p" | "q" | "r" | "s" | "t" | "u" | "v" | "w" | "x" | "y" | "z" | "A" | "B" | "C" | "D" | "E" | "F" | "G" | "H" | "I" | "J" | "K" | "L" | "M" | "N" | "O" | "P" | "Q" | "R" | "S" | "T" | "U" | "V" | "W" | "X" | "Y" | "Z" | "á" | "â" | "â" | "â" | "ê" | "ê" | "ê" | "ê" | "ê" | "ô" | "ô" | "ô" | "û" | "û" | "û" | "û" | "î" | "î" | "î" | "ï"




13.2.7 Leesteken

<leesteken> ::= "," | "." | "!" | "?" | ":" | ";" | "(" | ")" | "-"



13.2.8 Karakterreeks



Let op: zoals hierboven is te zien, kunnen in RegelSpraak Unicode karakters gebruikt worden. Unicode omvat (in 2023) bijna 150 duizend karakters. Deze karakters zijn vanwege de omvang niet volledig uitgewerkt in de syntax specificaties.


13.2.9 Lidwoord

 $\verb| <lidwoord> ::= <| bepaaldlidwoord> | <| onbepaaldlidwoord> |$



13.2.10 Bepaallidwoord

<bepaaldlidwoord> ::= "de" | "het"




13.2.11 Onbepaaldlidwoord

<onbepaaldlidwoord> ::= "een"



13.2.13 Waarde

<waarde> ::= <tekstwaarde> | <booleanwaarde> | <getalwaarde> | <dedato>



13.2.13 Enumeratiewaarde

<enumeratiewaarde> ::= "'" <karakterreeks> "\"



13.2.14 Tekstwaarde

<tekstwaarde> ::= "\"" <karakterreeks> "\""




13.2.15 Boolean waarde

<booleanwaarde> ::= ("waar" | "onwaar")



13.2.16 Getalwaarde

<getalwaarde> ::= <getal> [(<eenheidsafkorting>+) | (<eenheidsafkorting>+ "/"
<eenheidsafkorting>+)]




13.2.17 Percentage

<percentage> ::= <getal> "%"



13.2.18 De dato

<dedato> ::= "dd. "<datumwaarde> [<tijdwaarde>]



13.2.19 Datumwaarde

<datumwaarde> ::= <dag>"-"<maand>"-"<jaar>




13.2.20 Tijdwaarde

<tijdwaarde> ::= <uur>":"<minuut>":"<seconde>"."<milliseconde>




13.2.21 Dag

```
<dag> ::= "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" | "10" | "11" | "12" | "13" |
"14" | "15" | "16" | "17" | "18" | "19" | "20" | "21" | "22" | "23" | "24" | "25" | "26" |
"27" | "28" | "29" | "30" | "31"
```




13.2.22 Maand

<maand> ::= "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" | "10" | "11" | "12"




13.2.23 Jaar

<jaar> ::= ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9")
("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9")




13.2.24 Uur

<uur> ::= "00" | "01" | "02" | "03" | "04" | "05" | "06" | "07" | "08" | "09" | "10" | "11" |
"12" | "13" | "14" | "15" | "16" | "17" | "18" | "19" | "20" | "21" | "22" | "23"




13.2.25 Minuut

<minuut> ::= ("0" | "1" | "2" | "3" | "4" | "5") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9")




13.2.26 Seconde

```
<seconde> ::= ("0" | "1" | "2" | "3" | "4" | "5") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9")
```




13.2.27 Milliseconde

<milliseconde> ::= ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9") ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9")



13.2.28 Naamwoord

<naamwoord> ::= [<bepaaldlidwoord>] <naam> ["(mv:" <meervoudsvorm> ")"]



13.2.29 Naam

<naam> ::= <karakterreeks>



13.2.30 Meervoudsvorm

<meervoudsvorm> ::= <karakterreeks>




13.3 Objecten en parameters

13.3.1 Objecttypen


13.3.1.1 Objecttypedefinitie

<objecttypedefinitie> ::= "Objecttype" <naamwoord> ["(bezield)"] \n
((<koptekst> | <kenmerk> | <attribuut>) \n)+



13.3.1.2 Objecttype met lidwoord

<objecttypemetlidwoord> ::= <bepaaldlidwoord> <objecttypenaam>



13.3.1.3 Objecttypenaam

<objecttypenaam> ::= <karakterreeks>




13.3.1.4 Koptekst

<koptekst> ::= "---" <karakterreeks>



13.3.2 Attributen en kenmerken

13.3.2.1 Kenmerk



13.3.2.2 Bezittelijk kenmerk

<bezittelijkkenmerk> ::= <naamwoord> "kenmerk (bezittelijk)"



13.3.2.3 Bijvoeglijk kenmerk

<bijvoeglijkkenmerk> ::= "is" <naam> ["(mv: " <meervoudsvorm> ")"] "kenmerk (bijvoeglijk)"




13.3.2.4 Kenmerknaam

<kenmerknaam> ::= <karakterreeks>




13.3.2.5 Attribuut

<attribuut> ::= <naamwoord> \t (<datatype> | <domeinnaam>) ["gedimensioneerd met"
<dimensienaam>] ";"



13.3.2.6 Attribuut met lidwoord

<attribuutmetlidwoord> ::= [<bepaaldlidwoord>] <attribuutnaam>



13.3.2.7 Attribuutnaam


<attribuutnaam> ::= <karakterreeks>



13.3.3 Datatypen


13.3.3.1 Datatype

<datatype> ::= <numeriekdatatype> | <percentagedatatype> | <tekstdatatype> | <booleandatatype> | <datumtijddatatype>



13.3.3.2 Numeriek datatype

<numeriekdatatype> ::= "Numeriek (" <getalspecificatie> ")" ["met eenheid" [(<eenheidmacht>+
| "1") "/"] (<eenheidmacht>+)]



13.3.3.3 Percentage datatype

<percentagedatatype> ::= "Percentage (" <getalspecificatie> ")" ["met eenheid %" ["/"
<eenheidsafkorting>]]




13.3.3.4 Tekst datatype

<tekstdatatype> ::= "Tekst"



13.3.3.5 Boolean datatype

<booleandatatype> ::= "Boolean"




13.3.3.6 Datum-tijd datatype

<datumtijddatatype> ::= "Datum in dagen" | "Datum en tijd in millisecondes"



13.3.3.7 Getalspecificatie

<getalspecificatie> ::= ["negatief" | "niet-negatief" | "positief"] ("geheel getal" | "getal
met " <aantaldecimalen> " decimalen" | "getal")



13.3.3.8 Aantal decimalen

<aantaldecimalen> ::= <positiefgeheelgetal>



13.3.4 Domeinen


13.3.4.1 Domeindefinitie

<domeindefinitie> ::= "Domein" <domeinnaam> "is van het type" (<datatype> |
<enumeratiespecificatie>)



13.3.4.2 Enumeratiespecificatie

<enumeratiespecificatie> ::= "Enumeratie" \n (\t <enumeratiewaarde> \n)+



13.3.4.3 Domeinnaam

<domeinnaam> ::= <karakterreeks>



13.3.4.4 Enumeratiewaarde

<enumeratiewaarde> ::= <karakterreeks>




13.3.5 Eenheden

13.3.5.1 Eenheidsysteem



13.3.5.2 Omrekenspecificatie

 $\verb|\complex| convex in the convex of the co$



13.3.5.3 Eenheidsysteemnaam

<eenheidsysteemnaam> ::= <karakterreeks>




13.3.5.4 Eenheidsafkorting

<eenheidsafkorting> ::= <karakterreeks>



13.3.5.5 Eenheidmacht

 $\verb| <eenheid macht> ::= <eenheid safkorting>[^(<exponent>)]| \\$



13.3.5.6 Exponent

<exponent> ::= <geheelgetal>



13.3.6 Tijdlijnen

13.3.6.1 Tijdlijn

<tijdlijn> ::= "voor" ("elke dag" | "elke maand" | "elk jaar")



13.3.7 Dimensies


13.3.7.1 Dimensie

<dimensie> ::= "Dimensie" <bepaaldlidwoord> <dimensienaam> ", bestaande uit de "
<dimensienaammeervoud> <voorzetselspecificatie> \n (<labelwaardespecificatie> \n)+



13.3.7.2 Voorzetselspecificatie

```
<voorzetselspecificatie> ::= ("(na het attribuut met voorzetsel" ( "van" | "in" | "voor" |
"over" | "op" | "bij" | "uit" ) "):" | "(voor het attribuut zonder voorzetsel):")
```



13.3.7.3 Dimensienaam

<dimensienaam> ::= <karakterreeks>




13.3.7.4 Dimensienaam meervoud

<dimensienaammeervoud> ::= <karakterreeks>



13.3.7.5 Labelwaardespecificatie

<labelwaardespecificatie> ::= <digit>+". " <dimensiewaarde>



13.3.7.6 Dimensiewaarde


<dre><dimensiewaarde> ::= <karakterreeks>



13.3.8 Parameters

13.3.7.1 Parameterdefinitie

<parameterdefinitie> ::= "Parameter" <parametermetlidwoord> ":" (<datatype> | <domeinnaam>)



13.3.8.2 Parameter met lidwoord

<parametermetlidwoord> ::= <bepaaldlidwoord> <parameternaam>



13.3.8.3 Parameternaam


<parameternaam> ::= <karakterreeks>



13.3.9 Feittypen

13.3.9.1 Feittype definitie

<feittypedefinitie> ::= "Feittype" <feittypenaam> \n [<bepaaldlidwoord>] <rolnaam> ["(mv: "
<meervoudrolnaam> ")"] \t <objecttypenaam> \n [<bepaaldlidwoord>] <rolnaam> ["(mv: "
<meervoudrolnaam> ")"] \t <objecttypenaam> \n ("één" <rolnaam> | "meerdere"
<meervoudrolnaam>) <relatiebeschrijving> ("één" <rolnaam> | "meerdere" <meervoudrolnaam>)



13.3.9.2 Wederkerig feittype definitie

</rd>


</rd>

</rd>

</rd>

</rd>

<pre



13.3.9.3 Feittypenaam

<feittypenaam> ::= <karakterreeks>



13.3.9.4 Rolnaam

<rolnaam> ::= <karakterreeks>



13.3.9.5 Meervoudrolnaam

<meervoudrolnaam> ::= <karakterreeks>



13.3.9.6 Relatiebeschrijving

<relatiebeschrijving> ::= <karakterreeks>



13.3.10 Dagsoort

13.3.10.1 Dagsoort

<dagsoort> ::= "Dagsoort" <naamwoord>



13.3.10.2 Dagsoortnaam

<dagsoortnaam> ::= <karakterreeks>




13.4 RegelSpraak

13.4.1 Onderwerpketen

13.4.1.1 Onderwerpketen


<onderwerpketen> ::= ((<lidwoord> | "zijn") (<objecttypenaam> | <rolnaam> | <kenmerknaam>)) |
((<selector> "van" <onderwerpketen>) | <subselectie>)



13.4.2 RegelSpraak-regel

13.4.2.1 Regel

<regel> ::= "Regel" <regelnaam> (\n <regelversie>)+



13.4.2.2 Regelnaam

<regelnaam> ::= <karakterreeks>



13.4.2.3 Regelversie

<regelversie> ::= <versie> \n <regelspraakregel>




13.4.2.4 Versie

<versie> ::= "geldig" <versiegeldigheid>



13.4.2.5 Versiegeldigheid

<versiegeldigheid> ::= "altijd" | ("vanaf " (<datumwaarde> | <jaar>) ["t/m " (<datumwaarde> |
<jaar>)]) | ("t/m " (<datumwaarde> | <jaar>))




13.4.2.6 RegelSpraakregel

 $$$ \ensuremath{$<$ regelSpraakregel> ::= \ensuremath{$<$ regultantdeel> $$ is $$ i$




13.4.2.7 Selector

<selector> ::= [<lidwoord>] <rolnaam>



13.4.2.8 Subselectie

<subselectie> ::= <onderwerpketen> ("die" | "dat") <predicaat>




13.4.2.9 Attribuut van onderwerp

<attribuutvanonderwerp> ::= [<kwantificatie>] <attribuutmetlidwoord> "van" <onderwerpketen>



13.4.2.10 Variabelendeel

<variabelendeel> ::= "Daarbij geldt:" (\n \t <variabeleonderdeel>)* "."



13.4.2.11 Variabele onderdeel

<variabeleonderdeel> ::= [<bepaaldlidwoord>] <variabelenaam> "is" <expressie>



13.4.2.12 Variabelenaam


<variabelenaam> ::= <karakterreeks>



13.4.3 Resultaatdeel

13.4.3.1 Resultaatdeel


<resultaatdeel> ::= <gelijkstelling> | <kenmerktoekenning> | <objectcreatie> | <feitcreatie> |
<consistentieregel> | <initialisatie> | <verdeling> | <dagsoortdefinitie>



13.4.4 Gelijkstelling

13.4.4.1 Gelijkstelling

<gelijkstelling> := (<gelijkstellingtoekenning> | <gelijkstellingberekening>)



13.4.4.2 Gelijkstellingtoekenning

<gelijkstellingtoekenning> ::= <attribuutvanonderwerp> "moet gesteld worden op" <expressie>



13.4.4.3 Gelijkstellingberekening


<gelijkstellingberekening> ::= <attribuutvanonderwerp> "moet berekend worden als"
(<getalexpressie> | <datumexpressie>)



13.4.5 Kenmerktoekenning

13.4.5.1 Kenmerktoekenning


<kenmerktoekenning> ::= <onderwerpketen> ("is" | "heeft") ["een"] <kenmerknaam>



13.4.6 ObjectCreatie


13.4.6.1 Objectcreatie

<objectcreatie> ::= "Een" <onderwerpketen> "heeft een" <rolnaam> ["met" <waardetoekenning>
[("," <waardetoekenning>) * "en" <waardetoekenning>]]



13.4.6.2 Waarde toekenning

<waardetoekenning> ::= <attribuutwaardetoekenning> | <kenmerkwaardetoekenning>



13.4.6.3 Attribuutwaarde toekenning

<attribuutwaardetoekenning> ::= <attribuut> "gelijk aan" <expressie>



13.4.6.4 Kenmerkwaarde toekenning

<kenmerkwaardetoekenning> ::= <kenmerknaam> "gelijk aan" ("waar" | "onwaar")



13.4.7 FeitCreatie

13.4.7.1 Feitcreatie


<feitcreatie> ::= "Een" <rolnaam> "van een" <onderwerpketen> "is een" <rolnaam> "van een" <onderwerpketen>



13.4.8 Consistentieregels


13.4.8.1 Consistentieregel

<consistentieregel> := <enkelvoudigeconsistentieregel | <toplevelsamengesteldcriterium> |
<uniciteitscontrole>




13.4.8.2 Enkelvoudige consistentieregel

<enkelvoudigeconsistentieregel> ::= <getalconsistentie> | <datumconsistentie> |
<tekstconsistentie> | <objectconsistentie>



13.4.8.3 Getalconsistentie

<getalconsistentie> ::= <getalexpressie> "moet"
(<topleveleenzijdigegetalvergelijkingsoperatormeervoud> |
<topleveltweezijdigegetalvergelijkingsoperatormeervoud>)



13.4.8.4 Datumconsistentie

<datumconsistentie> ::= <datumexpressie> "moet"
(<topleveleenzijdigedatumvergelijkingsoperatormeervoud> |
<topleveltweezijdigedatumvergelijkingsoperatormeervoud>)




13.4.8.5 Tekstconsistentie

<tekstconsistentie> ::= <tekstexpressie> "moet"
(<topleveleenzijdigetekstvergelijkingsoperatormeervoud> |
topleveltweezijdigetekstvergelijkingsoperatormeervoud>)




13.4.8.6 Objectconsistentie

<objectconsistentie> ::= <objectexpressie> "moet"
(<topleveleenzijdigeobjectvergelijkingsoperatormeervoud> |
<topleveltweezijdigeobjectvergelijkingsoperatormeervoud>)




13.4.8.7 Toplevel samengesteld criterium

<toplevelsamengesteldcriterium> ::= "er moet worden voldaan aan" ("het volgende criterium:" |
(<consistentiekwantificatie> "volgende criteria:") <samengesteldcriteriumonderdeel>




13.4.8.8 Samengesteld criterium onderdeel

<samengesteldcriteriumonderdeel> ::= \n \t <genestcriterium> (\n <genestcriterium>)+




13.4.8.9 Genest criterium

<genestcriterium> ::= ("•")+ (<voorwaardevergelijking> | <samengesteldcriterium>)




13.4.8.10 Samengesteld criterium

<samengesteldcriterium> ::= "er wordt voldaan aan" ("het volgende criterium:" |
(<consistentiekwantificatie> "volgende criteria:") <samengesteldcriteriumonderdeel>




13.4.8.11 Uniciteitscontrole

<uniciteitscontrole> := (<alleattribuutvanonderwerp> | <uniciteitconcatenatie>) <vereniging>*
"moeten uniek zijn."




13.4.8.12 Vereniging

<vereniging> ::= "verenigd met" (<alleattribuutvanonderwerp> | <uniciteitconcatenatie>)




13.4.8.13 Alle attribuut van onderwerp

<alleattribuutvanonderwerp> ::= "de" <meervoudsvorm> "van alle" ((<objecttypenaam> |
<rolnaam>) "van" <onderwerpketen>) ["van" <onderwerpketen>]



13.4.8.14 Uniciteitsconcatenatie

<uniciteitconcatenatie> ::= "de concatenatie van" <meervoudsvorm> ("," <meervoudsvorm>) * "en"
<meervoudsvorm> "van alle" ((<objecttypenaam> | <rolnaam>) "van" <onderwerpketen>) ["van"
<onderwerpketen>]



13.4.9 Initialisatie

13.4.9.1 Initialisatie

<initialisatie> ::= <attribuutvanonderwerp> "moet geïnitialiseerd worden op" <expressie>




13.4.10 Verdeling

Opmerkingen:

- * Als <maximumaanspraak> of <verdeelafronding> worden gebruikt, dan is <onverdeelderest> verplicht.
- * <maximumaanspraak> kan alleen worden gebruikt als <verdelenzondergroepen> of <criteriumbijgelijkevolgorde> gelijk zijn aan "naar rato van".


13.4.10.1 Verdeling

<verdeling> ::= <attribuutvanonderwerp> "wordt verdeeld over" <attribuutvanonderwerp> ",
waarbij wordt verdeeld" (<verdelenzondergroepen> | <meervoudigcriterium>)




13.4.10.2 Verdelen zonder groepen

<verdelenzondergroepen> := "in gelijke delen" | ("naar rato van" <attribuutmetlidwoord>)



13.4.10.3 Meervoudig criterium

<meervoudigcriterium> ::= ":" \n (<verdelenovergroepen> | (<verdelenzondergroepen> ",")) [\n <maximumaanspraak>] [\n <veldeelafronding>] [\n <onverdeelderest>]



13.4.10.4 Verdelen over groepen

<verdelenovergroepen> ::= "- op volgorde van" (afnemende | toenemende) <attribuutmetlidwoord>
\n <criteriumbijgelijkevolgorde> ","




13.4.10.5 Criterium bij gelijke volgorde

<criteriumbijgelijkevolgorde> := "- bij even groot criterium" ("in gelijke delen" | ("naar rato van" <attribuutmetlidwoord>)) ","




13.4.10.6 Maximum aanspraak

<maximumaanspraak> ::= "- met een maximum van" <attribuutmetlidwoord> ","



13.4.10.7 Verdeelafronding

<verdeelafronding> ::= "- afgerond op" <geheelgetal> "decimalen naar beneden."



13.4.10.8 Onverdeelde rest

<onverdeelderest> ::= "Als onverdeelde rest blijft" <attribuutvanonderwerp> "over."



13.4.11 Dagsoortdefinitie

13.4.11.1 Dagsoortdefinitie


<dagsoortdefinitie> ::= "Een dag is een" <dagsoortnaam>



13.4.12 Voorwaardendeel

13.4.12.1 Voorwaardendeel

<voorwaardendeel> ::= "indien" (<toplevelelementairevoorwaarde> |
<toplevelsamengesteldevoorwaarde>)




13.4.12.2 Predicaat




13.4.12.3 Elementair predicaat

 $\begin{tabular}{ll} $< elementair predicaat > := < getal predicaat > | < tekst predicaat > | < datum predicaat > | < object predicaat >$




13.4.12.4 Samengesteld predicaat

<samengesteldpredicaat> ::= "aan" <kwantificatie> "volgende voorwaarde"["n"]" voldoen:"
(<samengesteldevoorwaardeonderdeel> | <toplevelvoorwaardevergelijking>)



13.4.12.5 Getalpredicaat

 $\verb| <getalpredicaat> := <topleveltweezijdigegetalvergelijkingsoperatormeervoud> <getalexpressie> | <getalex$



13.4.12.6 Tekstpredicaat

<tekstpredicaat> ::= <topleveltweezijdigetekstvergelijkingsoperatormeervoud> <tekstexpressie>



13.4.12.7 Datumpredicaat

<datumpredicaat> ::= <topleveltweezijdigedatumvergelijkingsoperatormeervoud> <datumexpressie>



13.4.12.8 Objectpredicaat


<objectpredicaat> ::= <topleveltweezijdigeobjectvergelijkingsoperatormeervoud>
<objectexpressie>



13.4.13 Samengestelde voorwaarde


13.4.13.1 Toplevel samengestelde voorwaarde

<toplevelsamengesteldevoorwaarde> ::= (<objectexpressie> | <referentie> | <aggregatie> | "er")
"aan" <voorwaardekwantificatie> "volgende voorwaarde"["n"] ("voldoet" | "voldoen" | "wordt
voldaan") ":" <samengesteldevoorwaardeonderdeel>




13.4.13.2 Geneste samengestelde voorwaarde

<genestesamengesteldevoorwaarde> ::= (<objectexpressie> | <referentie> | <aggregatie> | "er")
("voldoet" | "voldoen" | "wordt voldaan") "aan" <voorwaardekwantificatie> "volgende
voorwaarde"["n"]":" <samengesteldevoorwaardeonderdeel>



13.4.13.3 Consistentiekwantificatie


<consistentiekwantificatie> ::= "alle" | "geen van de" | (("ten minste" | "ten hoogste" |
"precies") (<getal> | "één" | "twee" | "drie" | "vier") "van de")



samengesteldevoorwaardeonderdeel


13.4.13.4 Voorwaardekwantificatie

<voorwaardekwantificatie> ::= "de" | "alle" | "geen van de" | (("ten minste" | "ten hoogste" |
"precies") (<getal> | "één" | "twee" | "drie" | "vier") "van de")




13.4.13.5 Kwantificatie

<kwantificatie> ::= "de" | "alle" | "al" | "geen van de" | (("ten minste" | "ten hoogste" |
"precies") (<getal> | "één" | "twee" | "drie" | "vier") "van de")




13.4.13.6 Samengestelde voorwaarde onderdeel

 $< same ngestel devoor waarde onder deel > ::= \n \t < geneste voor waarde > (\n < geneste voor waarde >) + (\n < geneste v$



13.4.13.7 Geneste voorwaarde

<genestevoorwaarde> ::= ("•")+ (<elementairevoorwaarde> | <genestesamengesteldevoorwaarde>)



13.4.14 Elementaire voorwaarde


13.4.14.1 Toplevel elementaire voorwaarde

<toplevelelementairevoorwaarde> ::= <toplevelvoorwaardevergelijking> |
<consistentievoorwaarde>




13.4.14.2 Toplevel voorwaardevergelijking

<toplevelvoorwaardevergelijking> ::= <toplevelgetalvergelijking> |
<toplevelobjectvergelijking> | <topleveltekstvergelijking> | <topleveldatumvergelijking> |
<toplevelbooleanvergelijking>



13.4.14.3 Toplevel getalvergelijking


<toplevelgetalvergelijking> ::= <topleveleenzijdigegetalvergelijking> |
<topleveltweezijdigegetalvergelijking>



13.4.14.4 Toplevel eenzijdige getalvergelijking

<topleveleenzijdigegetalvergelijking> ::= <getalexpressie>

<topleveleenzijdigegetalvergelijkingsoperator>



13.4.14.5 Toplevel tweezijdige getalvergelijking


<topleveltweezijdigegetalvergelijking> ::= <getalexpressie>

<topleveltweezijdigegetalvergelijkingsoperator> <getalexpressie>



13.4.14.6 Toplevel datumvergelijking


<topleveldatumvergelijking> := <topleveleenzijdigedatumvergelijking> |
<topleveltweezijdigedatumvergelijking>



13.4.14.7 Toplevel eenzijdige datumvergelijking

<topleveleenzijdigedatumvergelijking> ::= <datumexpressie>

<topleveleenzijdigedatumvergelijkingsoperator>



13.4.14.8 Toplevel tweezijdige datumvergelijking

<topleveltweezijdigedatumvergelijking> ::= <datumexpressie>

<topleveltweezijdigedatumvergelijkingsoperator> <datumexpressie>



13.4.14.9 Toplevel tekstvergelijking

<topleveltekstvergelijking> ::= <topleveleenzijdigetekstvergelijking> |


<topleveltweezijdigetekstvergelijking>



13.4.14.10 Toplevel eenzijdige tekstvergelijking

<topleveleenzijdigetekstvergelijking> ::= <tekstexpressie>

<topleveleenzijdigetekstvergelijkingsoperator>



13.4.14.11 Toplevel tweezijdige tekstvergelijking


<topleveltweezijdigetekstvergelijking> ::= <tekstexpressie>

<topleveltweezijdigetekstvergelijkingsoperator> <tekstexpressie>



13.4.14.12 Toplevel booleanvergelijking

<toplevelbooleanvergelijking> ::= <topleveleenzijdigebooleanvergelijking> | <topleveltweezijdigebooleanvergelijking>



13.4.14.13 Toplevel eenzijdige booleanvergelijking

<topleveleenzijdigebooleanvergelijking> ::= <booleanexpressie>

<topleveleenzijdigebooleanvergelijkingsoperator>



13.4.14.14 Toplevel tweezijdige booleanvergelijking

<topleveltweezijdigebooleanvergelijking> ::= <booleanexpressie>

<topleveltweezijdigebooleanvergelijkingsoperator> <booleanexpressie>



13.4.14.15 Toplevel objectvergelijking


<toplevelobjectvergelijking> ::= <topleveleenzijdigeobjectvergelijking> | <topleveltweezijdigeobjectvergelijking>



13.4.14.16 Toplevel eenzijdige objectvergelijking

<topleveleenzijdigeobjectvergelijking> ::= <objectexpressie>

<topleveleenzijdigeobjectvergelijkingsoperator>



13.4.14.17 Toplevel tweezijdige objectvergelijking


<topleveltweezijdigeobjectvergelijking> ::= (<objectexpressie> | <referentie>)

<topleveltweezijdigeobjectvergelijkingsoperator> <objectexpressie>



13.4.14.18 Consistentievoorwaarde

<consistentievoorwaarde> ::= "regelversie" <karakterreeks> "is" ("gevuurd" | "inconsistent")




13.4.14.19 Toplevel eenzijdige getalvergelijkingsoperator

<topleveleenzijdigegetalvergelijkingsoperator> ::=


<topleveleenzijdigegetalvergelijkingsoperatorenkelvoud> |

<topleveleenzijdigegetalvergelijkingsoperatormeervoud>




13.4.14.20 Toplevel eenzijdige getalvergelijkingsoperator enkelvoud

<topleveleenzijdigegetalvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("leeg is" | "gevuld is" | "aan de elfproef voldoet")




13.4.14.21 Toplevel eenzijdige getalvergelijkingsoperator meervoud

<topleveleenzijdigegetalvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("leeg zijn" | "gevuld zijn" | "aan de elfproef voldoen")



13.4.14.22 Toplevel tweezijdige getalvergelijkingsoperator

<topleveltweezijdigegetalvergelijkingsoperator> ::=
<topleveltweezijdigegetalvergelijkingsoperatorenkelvoud> |
<topleveltweezijdigegetalvergelijkingsoperatormeervoud>



13.4.14.23 Toplevel tweezijdige getalvergelijkingsoperator enkelvoud

<topleveltweezijdigegetalvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("gelijk is aan" | "ongelijk is aan" | "groter is dan" | "groter of gelijk is aan" | "kleiner
of gelijk is aan" | "kleiner is dan")




13.4.14.24 Toplevel tweezijdige getalvergelijkingsoperator meervoud

<topleveltweezijdigegetalvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("gelijk zijn aan" | "ongelijk zijn aan" | "groter zijn dan" | "groter of gelijk zijn aan" |
"kleiner of gelijk zijn aan" | "kleiner zijn dan")




13.4.14.25 Toplevel eenzijdige datumvergelijkingsoperator

<topleveleenzijdigedatumvergelijkingsoperator> ::=
<topleveleenzijdigedatumvergelijkingsoperatorenkelvoud> |
<topleveleenzijdigedatumvergelijkingsoperatormeervoud>




13.4.14.26 Toplevel eenzijdige datumvergelijkingsoperator enkelvoud

<topleveleenzijdigedatumvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("leeg is" | "gevuld is" | ("een" <dagsoort> "is"))



13.4.14.27 Toplevel eenzijdige datumvergelijkingsoperator meervoud

<topleveleenzijdigedatumvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("leeg zijn" | "gevuld zijn" | ("een" <dagsoort> "zijn"))



13.4.14.28 Toplevel tweezijdige datumvergelijkingsoperator

<topleveltweezijdigedatumvergelijkingsoperator> ::=
<topleveltweezijdigedatumvergelijkingsoperatorenkelvoud> |
<topleveltweezijdigedatumvergelijkingsoperatormeervoud>



13.4.14.29 Toplevel tweezijdige datumvergelijkingsoperator enkelvoud

<topleveltweezijdigedatumvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("gelijk is aan" | "ongelijk is aan" | "later is dan" | "later of gelijk is aan" | "eerder of
gelijk is aan" | "eerder is dan")



13.4.14.30 Toplevel tweezijdige datumvergelijkingsoperator meervoud

<topleveltweezijdigedatumvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("gelijk zijn aan" | "ongelijk zijn aan" | "later zijn dan" | "later of gelijk zijn aan" |
"eerder of gelijk zijn aan" | "eerder zijn dan")



13.4.14.31 Toplevel eenzijdige tekstvergelijkingsoperator

<topleveleenzijdigetekstvergelijkingsoperator> ::=
<topleveleenzijdigetekstvergelijkingsoperatorenkelvoud> |
<topleveleenzijdigetekstvergelijkingsoperatormeervoud>



13.4.14.32 Toplevel eenzijdige tekstvergelijkingsoperator enkelvoud

<topleveleenzijdigetekstvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("leeg is" | "gevuld is" | ("numeriek is met exact" <geheelgetal> "cijfers") | "aan de
elfproef voldoet")



13.4.14.33 Toplevel eenzijdige tekstvergelijkingsoperator meervoud

<topleveleenzijdigetekstvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("leeg zijn" | "gevuld zijn" | ("numeriek zijn met exact" <geheelgetal> "cijfers") | "aan de
elfproef voldoen")




13.4.14.34 Toplevel tweezijdige tekstvergelijkingsoperator

<topleveltweezijdigetekstvergelijkingsoperator> ::=


<topleveltweezijdigetekstvergelijkingsoperatorenkelvoud> |

<topleveltweezijdigetekstvergelijkingsoperatormeervoud>




13.4.14.35 Toplevel tweezijdige tekstvergelijkingsoperator enkelvoud

<topleveltweezijdigetekstvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("gelijk is aan" | "ongelijk is aan")



13.4.14.36 Toplevel tweezijdige tekstvergelijkingsoperator meervoud

<topleveltweezijdigetekstvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("gelijk zijn aan" | "ongelijk zijn aan")



13.4.14.37 Toplevel eenzijdige booleanvergelijkingsoperator

 $<\!\!\text{topleveleenzijdigebooleanvergelijkingsoperator>} ::=$


<topleveleenzijdigebooleanvergelijkingsoperatorenkelvoud> |

<topleveleenzijdigebooleanvergelijkingsoperatormeervoud>




13.4.14.38 Toplevel eenzijdige booleanvergelijkingsoperator enkelvoud

<topleveleenzijdigebooleanvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("leeg is" | "gevuld is")




13.4.14.39 Toplevel eenzijdige booleanvergelijkingsoperator meervoud

<topleveleenzijdigebooleanvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("leeg zijn" | "gevuld zijn")




13.4.14.40 Toplevel tweezijdige booleanvergelijkingsoperator

- <topleveltweezijdigebooleanvergelijkingsoperator> ::=
- <topleveltweezijdigebooleanvergelijkingsoperatorenkelvoud> |
- <topleveltweezijdigebooleanvergelijkingsoperatormeervoud>




13.4.14.41 Toplevel tweezijdige booleanvergelijkingsoperator enkelvoud

<topleveltweezijdigebooleanvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("gelijk is aan" | "ongelijk is aan")



13.4.14.42 Toplevel tweezijdige booleanvergelijkingsoperator meervoud


<topleveltweezijdigebooleanvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("gelijk zijn aan" | "ongelijk zijn aan")



13.4.14.43 Toplevel eenzijdige objectvergelijkingsoperator


<topleveleenzijdigeobjectvergelijkingsoperator> ::=

<topleveleenzijdigeobjectvergelijkingsoperatorenkelvoud> |
<topleveleenzijdigeobjectvergelijkingsoperatormeervoud>




13.4.14.44 Toplevel eenzijdige objectvergelijkingsoperator enkelvoud

<topleveleenzijdigeobjectvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
["een"] (<rolnaam> | <kenmerknaam>) ("is" | "heeft")



13.4.14.45 Toplevel eenzijdige objectvergelijkingsoperator meervoud

<topleveleenzijdigeobjectvergelijkingsoperatormeervoud> := [<geheleperiodevergelijking>]
["een"] (<rolnaam> | <kenmerknaam>) ("zijn" | "hebben")




13.4.14.46 Toplevel tweezijdige objectvergelijkingsoperator

<topleveltweezijdigeobjectvergelijkingsoperator> ::=
<topleveltweezijdigeobjectvergelijkingsoperatorenkelvoud> |
<topleveltweezijdigeobjectvergelijkingsoperatormeervoud>




13.4.14.47 Toplevel tweezijdige objectvergelijkingsoperator enkelvoud

<topleveltweezijdigeobjectvergelijkingsoperatorenkelvoud> ::= [<geheleperiodevergelijking>]
("gelijk is aan" | "ongelijk is aan")




13.4.14.48 Toplevel tweezijdige objectvergelijkingsoperator meervoud

<topleveltweezijdigeobjectvergelijkingsoperatormeervoud> ::= [<geheleperiodevergelijking>]
("gelijk zijn aan" | "ongelijk zijn aan")




13.4.14.49 Elementaire voorwaarde

<elementairevoorwaarde> ::= <voorwaardevergelijking> | <consistentievoorwaarde>



13.4.14.50 Voorwaardevergelijking

<voorwaardevergelijking> ::= <getalvergelijking> | <objectvergelijking> | <tekstvergelijking> | <datumvergelijking> | <booleanvergelijking> | <periodevergelijkingelementair>



13.4.14.51 Getalvergelijking

<getalvergelijking> ::= <eenzijdigegetalvergelijking> | <tweezijdigegetalvergelijking>



13.4.14.52 Eenzijdige getalvergelijking

<eenzijdigegetalvergelijking> := <getalexpressie> <eenzijdigegetalvergelijkingsoperator>



13.4.14.53 Tweezijdige getalvergelijking

<tweezijdigegetalvergelijking> ::= <getalexpressie> <tweezijdigegetalvergelijkingsoperator>
<getalexpressie>



13.4.14.54 Datumvergelijking

<datumvergelijking> ::= <eenzijdigedatumvergelijking> | <tweezijdigedatumvergelijking>



13.4.14.55 Eenzijdige datumvergelijking

<eenzijdigedatumvergelijking> ::= <datumexpressie> <eenzijdigedatumvergelijkingsoperator>



13.4.14.56 Tweezijdige datumvergelijking

<tweezijdigedatumvergelijking> ::= <datumexpressie> <tweezijdigedatumvergelijkingsoperator> <datumexpressie>



13.4.14.57 Tekstvergelijking

 $\verb|\display| < tweezijdigetekstvergelijking> | < tweezijdigetekstvergelij$



13.4.14.58 Eenzijdige tekstvergelijking

<eenzijdigetekstvergelijking> ::= <tekstexpressie> <eenzijdigetekstvergelijkingsoperator>




13.4.14.59 Tweezijdige tekstvergelijking

<tweezijdigetekstvergelijking> ::= <tekstexpressie> <tweezijdigetekstvergelijkingsoperator> <tekstexpressie>



13.4.14.60 Booleanvergelijking

<booleanvergelijking> ::= <eenzijdigebooleanvergelijking> | <tweezijdigebooleanvergelijking>



13.4.14.61 Eenzijdige booleanvergelijking

<eenzijdigebooleanvergelijking> ::= <booleanexpressie>

<eenzijdigebooleanvergelijkingsoperator>



13.4.14.62 Tweezijdige booleanvergelijking


<tweezijdigebooleanvergelijking> ::= <booleanexpressie>

<tweezijdigebooleanvergelijkingsoperator> <booleanexpressie>



13.4.14.63 Objectvergelijking

<objectvergelijking> ::= <eenzijdigeobjectvergelijking> | <tweezijdigeobjectvergelijking>



13.4.14.64 Eenzijdige objectvergelijking

<eenzijdigeobjectvergelijking> ::= <objectexpressie> <eenzijdigeobjectvergelijkingsoperator>



13.4.14.65 Tweezijdige objectvergelijking


<tweezijdigeobjectvergelijking> ::= (<objectexpressie> | <referentie>)

<tweezijdigeobjectvergelijkingsoperator> <objectexpressie>




13.4.14.66 Consistentievoorwaarde

<consistentievoorwaarde> ::= "regelversie" <karakterreeks> "is" ("gevuurd" | "inconsistent")




13.4.14.67 Periodevergelijking

<periodevergelijking> ::= <periodevergelijkingenkelvoudig> | <periodevergelijkingelementair>



13.4.14.68 Periodevergelijking enkelvoudig

<periodevergelijkingenkelvoudig> ::= ("vanaf" <datumexpressie>) | ("van" <datumexpressie>
"tot" <datumexpressie>) | ("van" <datumexpressie> "tot en met" <datumexpressie>) | ("tot"
<datumexpressie>) | ("tot en met" <datumexpressie>)




13.4.14.69 Periodevergelijking elementair

<periodevergelijkingelementair> ::= "het is de periode" <periodevergelijkingenkelvoudig>




13.4.14.70 Gehele periodevergelijking

<geheleperiodevergelijking> ::= ["niet"] "gedurende" ("het gehele jaar" | "de gehele
maand")




13.4.14.71 Eenzijdige getalvergelijkingsoperator

<eenzijdigegetalvergelijkingsoperator> ::= <eenzijdigegetalvergelijkingsoperatorenkelvoud> |
<eenzijdigegetalvergelijkingsoperatormeervoud>




13.4.14.72 Eenzijdige getalvergelijkingsoperator enkelvoud

<eenzijdigegetalvergelijkingsoperatorenkelvoud> ::= ("is" [<geheleperiodevergelijking>]
("leeg" | "gevuld")) | ("voldoet" [<geheleperiodevergelijking>] "aan de elfproef")




13.4.14.73 Eenzijdige getalvergelijkingsoperator meervoud

<eenzijdigegetalvergelijkingsoperatormeervoud> := ("zijn" [<geheleperiodevergelijking>]
("leeg" | "gevuld")) | ("voldoen" [<geheleperiodevergelijking>] "aan de elfproef")



13.4.14.74 Tweezijdige getalvergelijkingsoperator

<tweezijdigegetalvergelijkingsoperator> ::= <tweezijdigegetalvergelijkingsoperatorenkelvoud> |
<tweezijdigegetalvergelijkingsoperatormeervoud>




13.4.14.75 Tweezijdige getalvergelijkingsoperator enkelvoud

<tweezijdigegetalvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan" | "groter dan" | "groter of gelijk aan" | "kleiner of gelijk
aan" | "kleiner dan")




13.4.14.76 Tweezijdige getalvergelijkingsoperator meervoud

<tweezijdigegetalvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan" | "groter dan" | "groter of gelijk aan" | "kleiner of gelijk
aan" | "kleiner dan")




13.4.14.77 Eenzijdige datumvergelijkingsoperator

<eenzijdigedatumvergelijkingsoperator> ::= <eenzijdigedatumvergelijkingsoperatorenkelvoud> |
<eenzijdigedatumvergelijkingsoperatormeervoud>




13.4.14.78 Eenzijdige datumvergelijkingsoperator enkelvoud

<eenzijdigedatumvergelijkingsoperatorenkelvoud> ::= ("is" [<geheleperiodevergelijking>]
("leeg" | "gevuld")) | ([<geheleperiodevergelijking>] "een" <dagsoort> "is")




13.4.14.79 Eenzijdige datumvergelijkingsoperator meervoud

<eenzijdigedatumvergelijkingsoperatormeervoud> := ("zijn" [<geheleperiodevergelijking>]
("leeg" | "gevuld")) | ([<geheleperiodevergelijking>] "een" <dagsoort> "zijn")



13.4.14.80 Tweezijdige datumvergelijkingsoperator




13.4.14.81 Tweezijdige datumvergelijkingsoperator enkelvoud

<tweezijdigedatumvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan" | "later dan" | "later of gelijk aan" | "eerder of gelijk aan"
| "eerder dan")




13.4.14.82 Tweezijdige datumvergelijkingsoperator meervoud

<tweezijdigedatumvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan" | "later dan" | "later of gelijk aan" | "eerder of gelijk aan"
| "eerder dan")



13.4.14.83 Eenzijdige tekstvergelijkingsoperator

<eenzijdigetekstvergelijkingsoperator> ::= <eenzijdigetekstvergelijkingsoperatorenkelvoud> |
<eenzijdigetekstvergelijkingsoperatormeervoud>



13.4.14.84 Eenzijdige tekstvergelijkingsoperator enkelvoud

<eenzijdigetekstvergelijkingsoperatorenkelvoud> ::= ("is" [<geheleperiodevergelijking>]
("leeg" | "gevuld" | ("numeriek met exact" <geheelgetal> "cijfers"))) | ("voldoet"
[<geheleperiodevergelijking>] "aan de elfproef")




13.4.14.85 Eenzijdige tekstvergelijkingsoperator meervoud

<eenzijdigetekstvergelijkingsoperatormeervoud> := ("zijn" [<geheleperiodevergelijking>]
("leeg" | "gevuld" | ("numeriek met exact" <geheelgetal> "cijfers"))) | ("voldoen"
[<geheleperiodevergelijking>] "aan de elfproef")



13.4.14.86 Tweezijdige tekstvergelijkingsoperator



13.4.14.87 Tweezijdige tekstvergelijkingsoperator enkelvoud

<tweezijdigetekstvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")




13.4.14.88 Tweezijdige tekstvergelijkingsoperator meervoud

<tweezijdigetekstvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")



13.4.14.89 Eenzijdige booleanvergelijkingsoperator

<eenzijdigebooleanvergelijkingsoperator> ::= <eenzijdigebooleanvergelijkingsoperatorenkelvoud>
| <eenzijdigebooleanvergelijkingsoperatormeervoud>



13.4.14.90 Eenzijdige booleanvergelijkingsoperator enkelvoud

<eenzijdigebooleanvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("leeg" | "gevuld")



13.4.14.91 Eenzijdige booleanvergelijkingsoperator meervoud


<eenzijdigebooleanvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("leeg" | "gevuld")



13.4.14.92 Tweezijdige booleanvergelijkingsoperator


<tweezijdigebooleanvergelijkingsoperator> ::=
<tweezijdigebooleanvergelijkingsoperatorenkelvoud> |

<tweezijdigebooleanvergelijkingsoperatorenkelvoud>
<tweezijdigebooleanvergelijkingsoperatormeervoud>



13.4.14.93 Tweezijdige booleanvergelijkingsoperator enkelvoud

<tweezijdigebooleanvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")




13.4.14.94 Tweezijdige booleanvergelijkingsoperator meervoud

<tweezijdigebooleanvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")



13.4.14.95 Eenzijdige objectvergelijkingsoperator

<eenzijdigeobjectvergelijkingsoperator> ::= <eenzijdigeobjectvergelijkingsoperatorenkelvoud> |
<eenzijdigeobjectvergelijkingsoperatormeervoud>



13.4.14.96 Eenzijdige objectvergelijkingsoperator enkelvoud

<eenzijdigeobjectvergelijkingsoperatorenkelvoud> ::= ("is" | "heeft")
[<geheleperiodevergelijking>] ["een"] <kenmerknaam>




13.4.14.97 Eenzijdige objectvergelijkingsoperator meervoud

<eenzijdigeobjectvergelijkingsoperatormeervoud> ::= ("zijn" | "hebben")
[<geheleperiodevergelijking>] ["een"] <kenmerknaam>




13.4.14.98 Tweezijdige objectvergelijkingsoperator

<tweezijdigeobjectvergelijkingsoperator> ::= <tweezijdigeobjectvergelijkingsoperatorenkelvoud>
| <tweezijdigeobjectvergelijkingsoperatormeervoud>



13.4.14.99 Tweezijdige objectvergelijkingsoperator enkelvoud

<tweezijdigeobjectvergelijkingsoperatorenkelvoud> ::= "is" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")



13.4.14.100 Tweezijdige objectvergelijkingsoperator meervoud


<tweezijdigeobjectvergelijkingsoperatormeervoud> ::= "zijn" [<geheleperiodevergelijking>]
("gelijk aan" | "ongelijk aan")



13.4.15 Berekening

13.4.15.1 Berekening


<berekening> ::= <getalexpressie> ("plus" | "min" | "verminderd met" | "maal" | "gedeeld door"
| "gedeeld door (ABS)") <getalexpressie>



13.4.16 Expressie


13.4.16.1 Expressie

<expressie> ::= <getalexpressie> | <objectexpressie> | <datumexpressie> | <tekstexpressie> |
<booleanexpressie> | <expressietussenhaakjes> | <parametermetlidwoord> | <variabelenaam> |
<concatenatie>




13.4.16.2 Concatenatie

<concatenatie> ::= <expressie> [(", " <expressie>)*] (" en " | " of ") <expressie>




13.4.16.3 Getal expressie

<getalexpressie> := <begrenzingexpressie> | afrondingexpressie> | <getalfunctie> |
<getalaggregatie> | <attribuutvanonderwerp> | <getalwaarde> | <rekenjaar> | <jaaruitfunctie> |
<maanduitfunctie> | <daguitfunctie>




13.4.16.4 Datumexpressie

<datumexpressie> := <datumfunctie> | <attribuutvanonderwerp> | <dedato> | <datumaggregatie>




13.4.16.5 Objectexpressie

<objectexpressie> ::= [<kwantificatie>] <onderwerpketen>




13.4.16.6 Tekstexpressie

<tekstexpressie> ::= <tekstenwaardereeks> | <tekstwaarde> | <a href="fil



13.4.16.7 Boolean expressie

<booleanexpressie> ::= <booleanwaarde> | <attribuutvanonderwerp>




13.4.16.8 Expressie tussen haakjes

<expressietussenhaakjes> ::= "(" <expressie> ")"



13.4.16.9 Tekst en waardereeks

 $$$ \ensuremath{$<$} ::= "\"" (("«" \ensuremath{$<$} ">") | \ensuremath{$<$$} karakterreeks>) + "\"" ("" \ensuremath{$<$} ">"") | \ensuremath{$<$} karakterreeks>) + "\"" (""" \ensuremath{$<$} """ \ensuremath{$<$} """ \ensuremath{$<$} """ \ensuremath{$<$$} """ \ensuremath{$<$} """ \ensuremath{$<$$} """ \ensuremath{$<$} """ \ensuremath{$<$$} """ \ensu$




13.4.16.10 Functie

<functie> ::= <datumfunctie> | <getalfunctie>




13.4.16.11 Getalfunctie

<getalfunctie>::= <percentagefunctie> | <wortelfunctie> | <minimalewaardefunctie> |
<maximalewaardefunctie> | <verminderdmetfunctie> | <tijdsduurtussen>



13.4.16.12 Percentagefunctie

<percentagefunctie> ::= <getalexpressie> ["%"] "van" <getalexpressie>




13.4.16.13 Wortelfunctie

<wortelfunctie> ::= "de wortel van" <getalexpressie> <afronding>




13.4.16.14 Machtsverheffenfunctie

<machtsverheffenfunctie> ::= <getalexpressie> "tot de macht" <getalexpressie> <afronding>




13.4.16.15 Minimale waardefunctie

<minimalewaardefunctie> ::= "de minimale waarde van" <getalexpressie> (", " <getalexpressie>) *
"en " <getalexpressie>




13.4.16.16 Maximale waardefunctie

<maximalewaardefunctie> ::= "de maximale waarde van" <getalexpressie> (", " <getalexpressie>)*
"en " <getalexpressie>



13.4.16.17 Absolute waarde functie

<absolutewaardefunctie> ::= "de absolute waarde van (" <getalexpressie> ")"



13.4.16.18 Jaar uit functie

<jaaruitfunctie> ::= "het jaar uit" <datumexpressie>



13.4.16.19 Maand uit functie

<maanduitfunctie> ::= "de maand uit" <datumexpressie>



13.4.16.20 Dag uit functie

<daguitfunctie> ::= "de dag uit" <datumexpressie>




13.4.16.21 Afronding expressie

<afrondingexpressie> ::= <getalexpressie> <afronding>



13.4.16.22 Afronding

<afronding> ::= ("naar beneden" | "naar boven" | "rekenkundig" | "richting nul" | "weg van
nul") "afgerond op" <geheelgetal> "decimalen"




13.4.16.23 Begrenzing expressie

<begrenzingexpressie> ::= <getalexpressie> "," <begrenzing>



13.4.16.24 Begrenzing

<begrenzing> ::= (<begrenzingminimum> | <begrenzingmaximum> | <begrenzingmaximum>)



13.4.16.25 Begrenzingminimum

<begrenzingminimum> ::= "met een minimum van" <getalexpressie>



13.4.16.26 Begrenzingmaximum

<begrenzingmaximum> ::= "met een maximum van" <getalexpressie>




13.4.16.27 Rekenjaar

<rekenjaar> ::= "Rekenjaar"




13.4.16.28 Tijdsduur tussen

<tijdsduurtussen> ::= ("de tijdsduur van "| "de absolute tijdsduur van ") <datumexpressie> "tot" <datumexpressie> "in" ["hele"] <eenheidmeervoud>



13.4.16.29 Datumfunctie

<datumfunctie>::= <datummet> | <eerstepaasdagvan> | <dedato> | <datumberekening> | <eerstevan> | <laatstevan> | <rekendatum>



13.4.16.30 Rekendatum

<rekendatum> ::= "Rekendatum"



13.4.16.31 Datum met

<datummet> ::= "de datum met jaar, maand en dag(" <getalexpressie> ", " <getalexpressie> ", "




13.4.16.32 Eerste paasdag van

<eerstepaasdagvan> ::= "de eerste paasdag van (" <jaar> ")"




13.4.16.33 Datumberekening

<datumberekening> ::= <datumexpressie> ("plus" | "min") <getalexpressie> <eenheidsafkorting>




13.4.16.34 Eerste van

<eerstevan> ::= "de eerste van" <datumexpressie> (", " <datumexpressie>)* "en"
<datumexpressie>




13.4.16.35 Laatste van

<laatstevan> ::= "de laatste van" <datumexpressie> (", " <datumexpressie>) * "en"
<datumexpressie>




13.4.16.36 Referentie

<referentie> ::= <bezieldereferentie> | <nietbezieldereferentie> | <dagsoortreferentie>



13.4.16.37 Bezielde referentie

<bezieldereferentie> ::= "hij" | ("zijn" <attribuutnaam>) | ("zijn" <rolnaam>)



13.4.16.38 Niet bezielde referentie

<nietbezieldereferentie> ::= <objecttypemetlidwoord>




13.4.16.39 Dagsoort referentie

<dagsoortreferentie> ::= "de dag"




13.4.16.40 Aggregatie

<aggregatie> ::= <getalaggregatie> | <datumaggregatie> | <dimensieaggregatie>



13.4.16.41 Getalaggregatie

<getalaggregatie> ::= <getalaggregatiefunctie> <expressie>



13.4.16.42 Getalaggregatiefunctie

 $\$ <getalaggregatiefunctie> ::= "het aantal" | "de maximale waarde van" | "de minimale waarde van" | "de som van"



13.4.16.43 Datumaggregatie

<datumaggregatie> ::= <datumaggregatiefunctie> <expressie>




13.4.16.44 Datumaggregatiefunctie

<datumaggregatiefunctie> ::= "de eerste van" | "de laatste van"



13.4.16.45 Dimensieaggregatie

<dimensieaggregatie> ::= (<getalaggregatiefunctie> | <datumaggregatiefunctie>)
<attribuutvanonderwerp> <dimensieselectie>



13.4.16.46 Dimensieselectie

<dimensieselectie> ::= "over" (<aggregerenoveralledimensies> | <aggregerenoververzameling> |
<aggregerenoverbereik>)



13.4.16.47 Aggregatie over alle dimensies

<aggregerenoveralledimensies> ::= "alle" <dimensienaammeervoud>




13.4.16.48 Aggregatie over verzameling
<aggregate oververzameling ::= "de" <dimensienaammeervoud> "vanaf" <> "t/m" <>




13.4.16.49 Aggregeren over bereik

<aggregerenoverbereik> ::= "de" <dimensienaammeervoud> "in {" <dimensiewaarde> [(", " <dimensiewaarde>) * "en" <dimensiewaarde>] "}"




4.16.50 Conditie bij expressie

<conditiebijexpressie> ::= "gedurende de tijd dat" (<toplevelelementairevoorwaarde> | <toplevelsamengesteldevoorwaarde>) | <periodevergelijkingenkelvoudig>



4.16.50 Aggregeren over waarden per tijdseenheid

<waardepertijdseenheidaggregatie> ::= "het totaal van" <expressie> [<conditiebijexpressie>]



4.16.51 Telling aantal dagen

<tellingaantaldagen> ::= "het aantal dagen in" ("de maand" | "het jaar") "dat" <expressie>



4.16.52 Tijdsevenredig deel
<tijdsevenredigdeel> ::= "het tijdsevenredig deel per" ("maand" | "jaar") "van" <expressie>
<conditiebijexpressie>

